23. Запись с вариантной частью

0. Определение типа

В общем случае запись состоит из двух частей – постоянной и вариантной (рис. 23.1). В предыдущем разделе мы рассматривали запись только с постоянной частью, так как эта структура имеет наибольшее использование, чем запись с вариантной частью.

Рис. 23.1. – Тип запись

Вариантная часть записи позволяет хранить в записи разные поля (по именам, типам и количеству) в зависимости от значения ключа записи, который задается порядковым типом (ограниченное число значений). Перед типом ключа, который располагается после символа СASE, может находится имя поля, в котором будет храниться значение ключа. Если этого имени нет, то значение ключа в записи сохраняться не будет.

Таким образом, тип ключа задает количество различных вариантов записи. Вариант записи определяется ключом варианта (константа порядкового типа - типа ключа) и списком полей, входящих в данный вариант, не считая общих полей присущих всем вариантам и вынесенных в фиксированную часть записи. Список полей варианта заключается в круглые скобки и отделяется от ключа варианта двоеточием. Так как тип варианта, может включать значения, для которых нет полей, то в этом случае в круглых скобках список полей отсутствует (пустой вариант). Синтаксис определения вариантной части изображен на рис. 23.2.

Рис. 23.2. – Определение вариантной части

Все имена полей во всех вариантах должны быть уникальными, так как при выделении поля записи отнесение ее к  определенному варианту возможно только по имени поля. Список полей варианта по синтаксису совпадает с фиксированной частью записи (рис. 23.3).

Рис. 23.3. – Определение списка полей варианта

3. Структурная организация
(дополнения)

Объем, выделяемый под переменную типа, определяется вариантом, занимающим наибольших объем для хранения своих полей.

Например, для геометрических элементов необходимо иметь значение площади элемента на плоскости и величины, по которым вычисляется площадь. Пусть элементами будут - точка, квадрат, прямоугольник, треугольник и круг. Площадь точки всегда 0, площадь квадрата определяется через сторону (квадрат стороны), площадь прямоугольника определяется через величины двух смежных сторон (произведение длины на ширину), площадь треугольника можно определить по трем сторонам по формуле Герона (корень квадратный из произведения полупериметра и трех других сомножителей, определяемых как разность полупериметра и соответствующей стороны), площадь круга можно определить через радиус. Таким образом, общим полем будет площадь геометрического элемента, которое образует фиксированную часть записи. Если тип варианта поставить в соответствие набору геометрических элементов, то всего должно быть пять вариантов. В варианте точка нет никаких полей, в варианте квадрат - одно поле (сторона квадрата), в варианте прямоугольник - два поля (длина и ширина прямоугольника), в варианте треугольник - три поля (длины трех сторон), в варианте круг - одно поле (радиус окружности).

Общий объем структуры определяется объемом наибольшего варианта - в нашем случае это вариант треугольника. Если решим хранить в записи тип фигуры, указав имя поля ключа, то добавляется еще одно постоянное для всех типов элементов поле:

type geom_elem={перечень геометрических элементов}
               (tochka, kvadrat, priamougolnik, treugolnik, krug);

     inf_mod=record
               s:real; {постоянное поле - площадь}
               case tip {постоянное поле (поле ключа варианта)– тип фигуры}

                       : geom_elem of

                 tochka:();{вариант точка}

                 kvadrat,krug:(a:real);{варианты квадрат и круг}

                 priamougolnik:(l,h:real){вариант прямоугольник};

                 treugolnik:(l1,l2,l3:real){вариант треугольник}

             end;

Структура такой записи показана на рис. 23.4.

Рис. 23.4. – Структурная организация записи с полем ключа варианта

Если тип фигуры хранить в записи не будем, то тип записи определяется следующим образом:

type geom_elem={перечень геометрических элементов}
               (tochka, kvadrat, priamougolnik, treugolnik, krug);

     inf_mod=record

               s:real; {постоянное поле - площадь}
               case geom_elem of

                 tochka:();{вариант точка}

                 kvadrat,krug:(a:real);{варианты квадрат и круг}

                 priamougolnik:(l,h:real){вариант прямоугольник};

                 treugolnik:(l1,l2,l3:real){вариант треугольник}

             end;

Структура такой записи показана на рис. 23.5.

Рис. 23.5. – Структурная организация записи без поля ключа варианта

В этом случае, экономя память под запись, мы будем вынуждены в программе вводить дополнительные переменные, чтобы знать - информация о каком объекте представлена конкретной записью.

Пример программы

Разработаем алгоритм, вычисления площади геометрического элемента, используя структуру данных, определенную выше, как запись с вариантной частью и полем под ключ варианта:

program geom;

type geom_elem=(tochka, kvadrat, priamougolnik, treugolnik,

                krug);

     inf_mod=record

               s:real;

               case tip : geom_elem of

                 tochka:();

                 kvadrat,krug:(a:real);

                 priamougolnik:(l,h:real);

                 treugolnik:(l1,l2,l3:real)

             end;

var g_f:inf_mod; {конкретный геометрический элемент}

    cod_tip:0..4;{код типа геометрического элемента,

                  используется для ввода данных

                  перечисляемого типа}

    p:real;      {рабочая переменная - полупериметр

                   при вычислении площади треугольника}

begin

  writeln('Введите тип фигуры:');

  writeln('0 - точка;');

  writeln('1 - квадрат;');

  writeln('2 - прямоугольник;');

  writeln('3 - треугольник;');

  writeln('4 - круг.');

  readln(cod_tip);

  {определение типа геометрического элемента по номеру значения}

  g_f.tip:=geom_elem(cod_tip);

  {ввод исходных данных для геометрического элемента в соответствии с его типом}

  case g_f.tip of

  kvadrat:      begin

                  writeln('Сторона квадрата?');

                  readln(g_f.a);

                end;

  priamougolnik:begin

                  writeln('Стороны прямоугольника?');

                  readln(g_f.l, g_f.h);

                end;

  treugolnik:   begin

                  writeln('Стороны треугольника?');

                  readln( g_f.l1, g_f.l2, g_f.l3 );

                end;

  krug:         begin

                  writeln('Радиус окружности?');

                  readln(g_f.a);

                end

  end;{Сase}

  {вычисление площади геометрического элемента}

  case g_f.tip of

    tochka:       g_f.s:=0;

    kvadrat:      g_f.s:=sqr(g_f.a);

    priamougolnik:g_f.s:=g_f.l*g_f.h;

    treugolnik:   begin

                    p:=(g_f.l1+g_f.l2+g_f.l3)/2;

                    g_f.s:=sqrt(p*(p-g_f.l1)*(p-g_f.l2)*(p-g_f.l3));

                  end;

    krug:         g_f.s:=pi*sqr(g_f.a);

  end;{Сase}

  {вывод результата вычислений}

  write('Площадь ');

  case g_f.tip of

    tochka:       write('точки');

    kvadrat:      write('квадрата');

    priamougolnik:write('прямоугольника');

    treugolnik:   write('треугольника');

    krug:         write('круга');

  end;{Сase}

  writeln('=',g_f.s)

end.

